
This code provides important .examples of string (char array[]) manipulation crucial to your success in the upcomng RPNCalcualtor Project.
Royal St. George’s College		Design Engineering Studio
RPN CALCULATOR PREPARATION: (S)TRING EXERCISES
Advanced Computer Engineering School		ICS3U-E: AVR Foundations
Create an Arduino project called stringExercises and replace the default code with the sketch below.

// PROJECT  :stringExercises
// PURPOSE#1:Introduction of (Arduino) C string data type as array of char
// PURPOSE#2:Preparation for RPNCalculator Project
// COURSE   :ICS3U-E
// AUTHOR   :C. D'Arcy
// DATE     :2025 04 14
// MCU      :*
// STATUS   :Working
// REFERENCE:https://docs.arduino.cc/language-reference/en/variables/data-types/string/
//          :http://darcy.rsgc.on.ca/ACES/TEI3M/Tasks.html#RPN
//          :AVR-Libc:https://www.nongnu.org/avr-libc/user-manual/string_8h.html
//          :CPlusPlus: https://cplusplus.com/reference/cstring/
// NOTE     :No Third-Party Libraries

#define NULL '\0'  //good idea to name Control Characters
#define TAB  '\t'

//Various char array declaration from Reference #1 above
char str1[15];
char str2[8] = { 'a', 'r', 'd', 'u', 'i', 'n', 'o' };
char str3[8] = { 'a', 'r', 'd', 'u', 'i', 'n', 'o', NULL };
char str4[] = "arduino";
char str5[8] = "Arduino";
char str6[15] = "arduino";
char *p;  // pointer to a character
char newChar[2];

void setup() {
  Serial.begin(9600, SERIAL_8N1);  //2nd parameter is optional

  show(str1, sizeof(str1));  //confirm
  show(str2, sizeof(str2));  //
  show(str3, sizeof(str3));  //
  show(str4, sizeof(str4));  //
  show(str5, sizeof(str5));  //
  show(str6, sizeof(str6));  //

  // demonstrate similarity between char array[] and a 'pointer to a char'
  p = &str4[0];
  show(p, sizeof(str4));  //confirm
  show(p, sizeof(p));     //confirm
  Serial.println();

  



  strcpy(str5, "2");         //modify contents of a char array
  show(str5, sizeof(str5));  //confirm

  float operand = atof(str5);   //convert string to a float
  Serial.println(operand + 1);  //confirm

  strcat(str5, ".14159");       //join (concatenate) two strings
  operand = atof(str5);         //convert a char array of digits to a float
[bookmark: _GoBack]  Serial.println(operand + 1,5);//confirm

  char str7[8] = "17";       //demonstration of string modification
  show(str7, sizeof(str7));  //confirm
  str7[0] = '2';
  show(str7, sizeof(str7));  //confirm
  Serial.println(str7);      //confirm

  newChar[0] = '6';          //replace a character at a specific position
  strcat(str7, newChar);     //append char array to end of dest array
  show(str7, sizeof(str7));  //confirm
  Serial.println(str7);      //confirm
}
// Utility function to display the contents of a character array
void show(char a[], uint8_t length) {
  Serial.print(length);
  Serial.print(TAB);
  for (uint8_t i = 0; i < length; i++)
    Serial.print(a[i] == NULL ? '!' : a[i]);
  Serial.println();
}

void loop() {}

Serial Monitor Output:


15	!!!!!!!!!!!!!!!
8	arduino!
8	arduino!
8	arduino!
8	Arduino!
15	arduino!!!!!!!!
8	arduino!
2	ar

8	2!duino!
3.00
3.14159
8	17!!!!!!
8	27!!!!!!
27
8	276!!!!!
276

